


Pensieve header: A faster Jones polynomial program.

Based on <http://drorbn.net/syd3>

Setting Knots to contain all from Knot[3,1] to Knot[10,165] as well as GST48...



```

In[ ]:= KB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d},
  SetAttributes[p, Orderless];
  t1 = pd /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
  t2 = Expand[t1 /. PD -> Times];
  t3 = t2 //. {p[i_, j_] p[j_, k_] -> p[i, k]};
  t4 = t3 /. {p[i_, i_] -> d, p[i_, j_] ^2 -> d};
  Expand[t4 /. {B -> 1/A, d -> -A^2 - 1/A^2}]
]

In[ ]:= FKB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d, KB, todo, x},
  SetAttributes[p, Orderless];
  KB = 1;
  todo = pd;
  While[Length[todo] > 0,
    x = First[todo];
    todo = DeleteCases[todo, x];
    t1 = KB (x /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] * p[j, k]);
    t2 = Expand[t1];
    t3 = t2 //. {p[i_, j_] p[j_, k_] -> p[i, k]};
    t4 = t3 /. {p[i_, i_] -> d, p[i_, j_] ^2 -> d};
    KB = Expand[t4 /. {B -> 1/A, d -> -A^2 - 1/A^2}];
  ];
  KB
]

```

In[*]:= **FKB**[**Knot**[8, 17] /. **Knots**]

$$\text{Out[*]} = -\frac{1}{A^{18}} + \frac{2}{A^{14}} - \frac{2}{A^{10}} + \frac{1}{A^6} - \frac{1}{A^2} - A^2 + A^6 - 2A^{10} + 2A^{14} - A^{18}$$

In[*]:= **Timing**[**Table**[**FKB**[**Knot**[10, k] /. **Knots**], {k, 165}];]

Out[*]= {6.03125, Null}

```
In[*]:= ThinPosition[pd_PD] := Module[{todo, done, out, c},
  todo = List@@pd; done = {}; out = PD[];
  While[todo != {},
    AppendTo[out, c = RandomChoice@MaximalBy[todo, Length[done ∩ List@@ #] &]];
    todo = DeleteCases[todo, c];
    done = done ∪ List@@c;
  out
]
```

In[*]:= **EFKB**[*pd_PD*] := **FKB**[**ThinPosition**[*pd*]]

In[*]:= **Table**[**EFKB**[**Knot**[7, k] /. **Knots**] == **FKB**[**Knot**[7, k] /. **Knots**], {k, 7}]

Out[*]= {True, True, True, True, True, True, True}

In[*]:= **Timing**[**Table**[**EFKB**[**Knot**[10, k] /. **Knots**], {k, 165}];]

Out[*]= {1.75, Null}

In[*]:= **Timing**[**EFKB**[**GST48** /. **Knots**]]

$$\text{Out[*]} = \left\{ 6.60938, \frac{1}{A^{60}} - \frac{2}{A^{56}} + \frac{1}{A^{52}} + \frac{1}{A^{48}} - \frac{2}{A^{44}} - \frac{1}{A^{24}} + \frac{2}{A^{20}} + \frac{1}{A^{16}} - \frac{2}{A^{12}} + \frac{4}{A^8} - \frac{2}{A^4} - 2A^8 - 2A^{16} + A^{20} - A^{28} + A^{36} \right\}$$